



Tech Info Library

Applesoft Tutorial: Display problems on the Apple IIc

Revised: 10/8/84
Security: Everyone

Applesoft Tutorial: Display problems on the Apple IIc

=====
Customers are finding that some of the programs on the Applesoft Sampler diskette (included with the Applesoft Tutorial), when used on an Apple IIc, do not function as expected. Menus generated by these programs operate correctly when used on an Apple IIe, but there is a vertical displacement of one line between menu selection numbers and menu item descriptions when executed on an Apple IIc.

The problem resides in three programs (CONVERTER, MAGIC.MENU and DISK.MENU) located on the Applesoft Sampler diskette. Load the "CONVERTER" program and list line number 625. This line should currently read;

```
625 VTAB PEEK(37): IF COL80 THEN VTAB PEEK(1531)
```

Change this line to read:

```
625 IF COL80 THEN VTAB PEEK(1531) : GOTO 630
```

and then add the following line:

```
627 VTAB PEEK(37)
```

SAVE the file "CONVERTER".

Make the same changes to the files "DISK.MENU" and "MAGIC.MENU"

These changes will fix the problem.

We have been informed that these changes will soon be included in final product. However, the problem exists in product that is already shipping, so you can and should expect to hear from customers about this. Our suggestion is to perform the above changes to the CONVERTER MAGIC.MENU and DISK.MENU programs and give the fixed version of the program to those customers that need it. We expect to be informed when the change takes place in final (shipped) product.

As an added attraction we have found that two programs mentioned in the

manual are noyt included on the disk. The listings for MOUSE.MOVE and MOUSE.DRAW follow. Please distribute them to customers who did not recieve copies on their disks.

MOUSE.MOVE

```
10 HOME
20 PRINT "THIS IS A DEMONSTRATION OF THE MOUSE"
30 PRINT CHR$(4);"PR#4": PRINT CHR$(1)
40 PRINT CHR$(4);"PR#0"
50 PRINT CHR$(4);"IN#4"
60 INPUT ";"X,Y,S
70 VTAB 10: PRINT X;" ",Y;" ",S" "
80 IF S > 0 THEN 60
90 PRINT CHR$(4);"IN#0"
100 PRINT CHR$(4);"PR#4": PRINT CHR$(0)
110 PRINT CHR$(4);"PR#0"
120 POKE - 16368,0: REM CLEAR KEYBOARD STROBE
130 END
```

MOUSE.DRAW

```
10 REM USES MOUSE TO DRAW LOW-RES GRAPHICS
100 GOSUB 1000: REM TURN ON THE MOUSE
110 PRINT CHR$(4);"IN#4"
120 INPUT ";"X,Y,S: REM READ MOUSE POSITION
130 IF S = 1 THEN 100: REM CLEAR THE SCREEN
140 IF S < 0 THEN 300: REM TIME TO QUIT?
150 REM SCALE THE POSITION
160 X = INT (X / 25.575)
170 Y = INT (Y / 25.575)
180 PLOT X,Y
190 GOTO 120
300 REM CHECK IF TIME TO QUIT
310 POKE - 16368,0: REM CLEAR KEYBOARD STROBE
320 PRINT CHR$(4);"IN#0"
330 VTAB 22: PRINT "PRESS RETURN TO CONT OR ESC TO QUIT"
340 VTAB 22: HTAB 39: GET A$: PRINT
350 IF A$ = CHR$(13) THEN HOME : GOTO 110
360 IF A$ < > CHR$(27) THEN 340
370 REM CLEAR SCREEN AND TURN OFF MOUSE
380 TEXT : HOME
390 PRINT CHR$(4);"PR#4": PRINT CHR$(0)
400 PRINT CHR$(4);"PR#0"
410 END
1000 REM CLEAR SCREEN AND INITIALIZE MOUSE
1010 HOME : GR
1020 COLOR= 15
1030 PRINT CHR$(4);"PR#4": PRINT CHR$(1)
1040 PRINT CHR$(4);"PR#0"
1050 RETURN
```

Apple Technical Support

